

NAG Library Routine Document

C05AJF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

C05AJF attempts to locate a zero of a continuous function using a continuation method based on a secant iteration.

2 Specification

```
SUBROUTINE C05AJF(X, EPS, ETA, F, NFMAX, IFAIL)
INTEGER          NFMAX, IFAIL
double precision X, EPS, ETA, F
EXTERNAL         F
```

3 Description

C05AJF attempts to obtain an approximation to a simple zero α of the function $f(x)$ given an initial approximation x to α . The zero is found by a call to C05AXF whose specification should be consulted for details of the method used.

The approximation x to the zero α is determined so that at least one of the following criteria is satisfied:

- (i) $|x - \alpha| \sim \text{EPS}$,
- (ii) $|f(x)| < \text{ETA}$.

4 References

None.

5 Parameters

- 1: X – ***double precision*** *Input/Output*
On entry: an initial approximation to the zero.
On exit: the final approximation to the zero, unless $\text{IFAIL} = 1, 2$ or 5 on exit, in which case it contains no useful information.
- 2: EPS – ***double precision*** *Input*
On entry: an absolute tolerance to control the accuracy to which the zero is determined. In general, the smaller the value of EPS the more accurate X will be as an approximation to α . Indeed, for very small positive values of EPS , it is likely that the final approximation will satisfy $|X - \alpha| < \text{EPS}$. You are advised to call the routine with more than one value for EPS to check the accuracy obtained.
Constraint: $\text{EPS} > 0.0$.
- 3: ETA – ***double precision*** *Input*
On entry: a value such that if $|f(x)| < \text{ETA}$, x is accepted as the zero. ETA may be specified as 0.0 (see Section 7).

- 4: F – *double precision* FUNCTION, supplied by the user. *External Procedure*
 F must evaluate the function f whose zero is to be determined.

The specification of F is:

```
double precision FUNCTION F(XX)
double precision          XX
```

1: XX – *double precision* *Input*
On entry: the point at which the function must be evaluated.

F must be declared as EXTERNAL in the (sub)program from which C05AJF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

- 5: NFMAX – INTEGER *Input*

On entry: the maximum permitted number of calls to F from C05AJF. If F is inexpensive to evaluate, NFMAX should be given a large value (say > 1000).

Constraint: NFMAX > 0.

- 6: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, $\text{EPS} \leq 0.0$,
 or $\text{NFMAX} \leq 0$.

IFAIL = 2

An internally calculated scale factor has the wrong order of magnitude for the problem. If this error exit occurs, you are advised to call C05AXF instead where different scale values can be tried.

IFAIL = 3

Either the function $f(x)$ given by F has no zero near X or too much accuracy has been requested in calculating the zero. The first is a more likely cause of this error exit and you should check the coding of F and make an independent investigation of its behaviour near X. The second can be alleviated by increasing EPS.

IFAIL = 4

More than NFMAX calls have been made to F. This error exit can occur because NFMAX is too small for the problem (essentially because X is too far away from the zero) or for either of the reasons given under IFAIL = 3 above. If NFMAX is increased considerably and this error exit

occurs again at approximately the same final value of X , then it is likely that one of the reasons given under $IFAIL = 3$ is the cause.

$IFAIL = 5$ (C05AXF)

A serious error has occurred in the specified routine. Check all subroutine calls. Seek expert help.

7 Accuracy

The levels of accuracy depend on the values of EPS and ETA . If full machine accuracy is required, they may be set very small, resulting in an exit with $IFAIL = 3$ or 4 , although this may involve many more iterations than a lesser accuracy. You are recommended to set $ETA = 0.0$ and to use EPS to control the accuracy, unless you have considerable knowledge of the size of $f(x)$ for values of x near the zero.

8 Further Comments

The time taken by C05AJF depends primarily on the time spent evaluating the function f (see Section 5) and on how close the initial value of X is to the zero.

If a more flexible way of specifying the function f is required or if you wish to have closer control of the calculation, then the reverse communication routine C05AXF is recommended instead of C05AJF.

9 Example

This example calculates the zero of $f(x) = e^{-x} - x$ from a starting value $X = 1.0$. Two calculations are made with $EPS = 1.0D-3$ and $1.0D-4$ for comparison purposes, with $ETA = 0.0$ in both cases.

9.1 Program Text

```
*      C05AJF Example Program Text
*      Mark 14 Revised. NAG Copyright 1989.
*      .. Parameters ..
      INTEGER                NOUT
      PARAMETER              (NOUT=6)
*      .. Local Scalars ..
      DOUBLE PRECISION EPS, ETA, X
      INTEGER                IFAIL, K, NFMAX
*      .. External Functions ..
      DOUBLE PRECISION F
      EXTERNAL               F
*      .. External Subroutines ..
      EXTERNAL               C05AJF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'C05AJF Example Program Results'
      WRITE (NOUT,*)
      DO 20 K = 3, 4
         EPS = 10.0D0**(-K)
         X = 1.0D0
         ETA = 0.0D0
         NFMAX = 200
         IFAIL = 1

*
         CALL C05AJF(X, EPS, ETA, F, NFMAX, IFAIL)
*

         IF (IFAIL.EQ.0) THEN
            WRITE (NOUT,99998) 'With EPS = ', EPS, '    root = ', X
         ELSE IF (IFAIL.LT.0) THEN
            WRITE (NOUT,*)
            WRITE (NOUT,99999) ' ** C05AJF returned with IFAIL = ',
+              IFAIL
            GO TO 40
         ELSE
            WRITE (NOUT,99999) ' ** C05AJF returned with IFAIL = ',
+              IFAIL
            IF (IFAIL.EQ.3 .OR. IFAIL.EQ.4) THEN
```

```
                WRITE (NOUT,99998) 'With EPS = ', EPS, ' final value = ',
+                X
                END IF
            END IF
20 CONTINUE
40 CONTINUE
*
99999 FORMAT (1X,A,I5)
99998 FORMAT (1X,A,E10.2,A,F14.5)
END
*
DOUBLE PRECISION FUNCTION F(X)
*   .. Scalar Arguments ..
DOUBLE PRECISION X
*   .. Intrinsic Functions ..
INTRINSIC      EXP
*   .. Executable Statements ..
F = EXP(-X) - X
RETURN
END
```

9.2 Program Data

None.

9.3 Program Results

C05AJF Example Program Results

With EPS =	0.10E-02	root =	0.56715
With EPS =	0.10E-03	root =	0.56715
